
1. %ABS (Absolute Value of Expression)

     %ABS(numeric expression) 

%ABS returns the absolute value of the numeric expression specified as the parameter. If the value of the numeric expression is non-negative, the value is returned unchanged. If the value is negative, the value returned is the value of the expression but with the negative sign removed. 

%ABS may be used either in expressions or as parameters to keywords. When used with keywords, the operand must be a numeric literal, a constant name representing a numeric value, or a built-in function with a numeric value known at compile-time. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D f8              s              8f   inz (-1)                                                 | 

   |   D i10             s             10i 0 inz (-123)                                               | 

   |   D p7              s              7p 3 inz (-1234.567)                                          | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      f8  = %abs (f8)                                                  | 

   |   C                   eval      i10 = %abs (i10 - 321)                                           | 

   |   C                   eval      p7  = %abs (p7)                                                  | 

   |    * The value of "f8" is now 1.                                                                 | 

   |    * The value of "i10" is now 444.                                                              | 

   |    * The value of "p7" is now 1234.567.                                                          | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 133. %ABS Example 


2 %ADDR (Get Address of Variable)

     %ADDR(variable) 

     %ADDR(variable(index)) 

     %ADDR(variable(expression)) 

%ADDR returns a value of type basing pointer. The value is the address of the specified variable. It may only be compared with and assigned to items of type basing pointer. 

If %ADDR with an array index parameter is specified as parameter for definition specification keywords INZ or CONST, the array index must be known at compile-time. The index must be either a numeric literal or a numeric constant. 

In an EVAL operation where the result of the assignment is an array with no index, %ADDR on the right hand side of the assignment operator has a different meaning depending on the argument for the %ADDR. If the argument for %ADDR is an array name without an index and the result is an array name, each element of the result array will contain the address of the beginning of the argument array. If the argument for %ADDR is an array name with an index of (*), then each element of the result array will contain the address of the corresponding element in the argument array. This is illustrated in Figure 134. 

If the variable specified as parameter is a table, multiple occurrence data structure, or subfield of a multiple occurrence data structure, the address will be the address of the current table index or occurrence number. 

If the variable is based, %ADDR returns the value of the basing pointer for the variable. If the variable is a subfield of a based data structure, the value of %ADDR is the value of the basing pointer plus the offset of the subfield. 

If the variable is specified as a PARM of the *ENTRY PLIST, %ADDR returns the address passed to the program by the caller. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * The following set of definitions is valid since the array                                   | 

   |    * index has a compile-time value                                                              | 

   |    *                                                                                             | 

   |   D   ARRAY         S             20A   DIM (100)                                                | 

   |    * Set the pointer to the address of the seventh element of the array.                         | 

   |   D   PTR           S               *   INZ (%ADDR(ARRAY(SEVEN)))                                | 

   |   D   SEVEN         C                   CONST (7)                                                | 

   |    *                                                                                             | 

   |   D DS1             DS                  OCCURS (100)                                             | 

   |   D                               20A                                                            | 

   |   D   SUBF                        10A                                                            | 

   |   D                               30A                                                            | 

   |   D CHAR10          S             10A   BASED (P)                                                | 

   |   D PARRAY          S               *   DIM(100)                                                 | 

   |                                                                                                  | 

   |                                                                                                  | 

   |                                                                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |   C     23            OCCUR     DS1                                                              | 

   |   C                   EVAL      SUBF = *ALL'abcd'                                                | 

   |   C                   EVAL      P = %ADDR (SUBF)                                                 | 

   |   C                   IF        CHAR10 = SUBF                                                    | 

   |    *                     This condition is true                                                  | 

   |   C                   ENDIF                                                                      | 

   |   C                   IF        %ADDR (CHAR10) = %ADDR (SUBF)                                    | 

   |    *                     This condition is also true                                             | 

   |   C                   ENDIF                                                                      | 

   |    * The following statement also changes the value of SUBF                                      | 

   |   C                   EVAL      CHAR10 = *ALL'efgh'                                              | 

   |   C                   IF        CHAR10 = SUBF                                                    | 

   |    *                     This condition is still true                                            | 

   |   C                   ENDIF                                                                      | 

   |    *--------------------------------------------------------------                               | 

   |   C           24      OCCUR     DS1                                                              | 

   |   C                   IF        CHAR10 = SUBF                                                    | 

   |    *                     This condition is no longer true                                        | 

   |   C                   ENDIF                                                                      | 

   |    *--------------------------------------------------------------                               | 

   |    * The address of an array element is taken using an expression                                | 

   |    * as the array index                                                                          | 

   |    *                                                                                             | 

   |   C                   EVAL      P = %ADDR (ARRAY (X + 10))                                       | 

   |    *--------------------------------------------------------------                               | 

   |    * Each element of the array PARRAY contains the address of the                                | 

   |    * first element of the array ARRAY.                                                           | 

   |   C                   EVAL      PARRAY = %ADDR(ARRAY)                                            | 

   |    * Each element of the array PARRAY contains the address of the                                | 

   |    * corresponding element of the array ARRAY                                                    | 

   |   C                   EVAL      PARRAY = %ADDR(ARRAY(*))                                         | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 134. %ADDR Example 


3 %CHAR (Convert to Character Data)

     %CHAR(expression) 


| %CHAR converts the value of the expression from graphic, UCS-2, numeric, date, time or timestamp data to type character. The converted value remains unchanged, but is returned in a format that is compatible with character data. 


| If the parameter is a constant, the conversion will be done at compile 
| time. 


| If a UCS-2 conversion results in substitution characters, a warning 
| message will be given in the compiler listing if the parameter is a 
| constant. Otherwise, status 00050 will be set at run time but no error 
| message will be given. 

For graphic data, the value returned includes the shift-in and shift-out characters. For example, if a 5 character graphic field is coverted, the returned value is 12 characters (10 bytes of graphic data plus the two shift characters). If the value of the expression has a variable length, the value returned is in varying format. 

For date, time, or timestamp data, the returned value includes any 
| separator characters. The format and separators of the result are the same 
| as that of the parameter. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D ChineseName     S             20G   VARYING INZ(G'oXXYYZZi')                                 | 

   |   D date            S               D   INZ(D'1997/02/03')                                       | 

   |   D time            S               T   INZ(T'12:23:34')                                         | 

   |   D result          S            100A   VARYING                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   EVAL      result = 'It is ' + %CHAR(time)                                  | 

   |   C                                    + ' on ' + %CHAR(date)                                    | 

   |    * result = 'It is 12:23:34 on 1997/02/03'                                                     | 

   |    *                                                                                             | 

   |   C                   EVAL      result = 'The time is now '                                      | 

   |   C                                    + %SUBST(%CHAR(time):1:5) + '.'                           | 

   |    * result = 'The time is now 12:23.'                                                           | 

   |    *                                                                                             | 

   |   C                   EVAL      result = 'The customer''s name is '                              | 

   |   C                                    + %CHAR(ChineseName) + '.'                                | 

   |    * result = 'The customer's name is oXXYYZZi.'                                                 | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 135. %CHAR Example 


4 %DEC (Convert to Packed Decimal Format)

     %DEC(numeric expression{:precision:decimal places}) 

%DEC converts the value of the numeric expression to decimal (packed) format with precision digits and decimal places decimal positions. The precision and decimal places must be numeric literals, named constants that represent numeric literals, or built-in functions with a numeric value known at compile-time. 

Parameters precision and decimal places may be omitted if the type of numeric expression is not float. If these parameters are omitted, the precision and decimal places are taken from the attributes of numeric expression. 

Subtopics: 

· 4.1 %DECH (Convert to Packed Decimal Format with Half Adjust) 


4.1 %DECH (Convert to Packed Decimal Format with Half Adjust)

     %DECH(numeric expression :precision:decimal places ) 

%DECH is the same as %DEC except that if numeric expression is a decimal or float value, half adjust is applied to the value of numeric expression when converting to the desired precision. No message is issued if half adjust cannot be performed. 

Unlike, %DEC, all three parameters are required. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D p7              s              7p 3 inz (1234.567)                                           | 

   |   D s9              s              9s 5 inz (73.73442)                                           | 

   |   D f8              s              8f   inz (123.456789)                                         | 

   |   D result1         s             15p 5                                                          | 

   |   D result2         s             15p 5                                                          | 

   |   D result3         s             15p 5                                                          | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      result1 = %dec (p7) + 0.011                                      | 

   |   C                   eval      result2 = %dec (s9 : 5: 0)                                       | 

   |   C                   eval      result3 = %dech (f8: 5: 2)                                       | 

   |    * The value of "result1" is now 1234.57800.                                                   | 

   |    * The value of "result2" is now   73.00000                                                    | 

   |    * The value of "result3" is now  123.46000.                                                   | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 136. %DEC and %DECH Example 


5 %DECPOS (Get Number of Decimal Positions)

     %DECPOS(numeric expression) 

%DECPOS returns the number of decimal positions of the numeric variable or expression. The value returned is a constant, and so may participate in constant folding. 

The numeric expression must not be a float variable or expression. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D p7              s              7p 3 inz (8236.567)                                           | 

   |   D s9              s              9s 5 inz (23.73442)                                           | 

   |   D result1         s              5i 0                                                          | 

   |   D result2         s              5i 0                                                          | 

   |   D result3         s              5i 0                                                          | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      result1 = %decpos (p7)                                           | 

   |   C                   eval      result2 = %decpos (s9)                                           | 

   |   C                   eval      result3 = %decpos (p7 * s9)                                      | 

   |    * The value of "result1" is now  3.                                                           | 

   |    * The value of "result2" is now  5.                                                           | 

   |    * The value of "result3" is now  8.                                                           | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 137. %DECPOS Example 

See Figure 154 in topic 18.2 for an example of %DECPOS with %LEN. 




6 %DIV (Return Integer Portion of Quotient)

 |   %DIV(n:m) 


| %DIV returns the integer portion of the quotient that results from 
| dividing operands n by m. The two operands must be numeric values with 
| zero decimal positions. If either operand is a packed, zoned, or binary 
| numeric value, the result is packed numeric. If either operand is an 
| integer numeric value, the result is integer. Otherwise, the result is 
| unsigned numeric. Float numeric operands are not allowed. (See also "%REM 
| (Return Integer Remainder)" in topic 23.) 


| If the operands are constants that can fit in 8-byte integer or unsigned 
| fields, constant folding is applied to the built-in function. In this 
| case, the %DIV built-in function can be coded in the definition 
| specifications. 


| This function is illustrated in Figure 159 in topic 23. 


7 %EDITC (Edit Value Using an Editcode)

     %EDITC(numeric : editcode {: *ASTFILL | *CURSYM | currency-symbol}) 

This function returns a character result representing the numeric value edited according to the edit code. In general, the rules for the numeric value and edit code are identical to those for editing numeric values in output specifications. The third parameter is optional, and if specified, must be one of: 

*ASTFILL 

Indicates that asterisk protection is to be used. This means that leading zeros are replaced with asterisks in the returned 
| value. For example, %EDITC(-0012.5 : 'K' : *ASTFILL) returns '***12.5-'. 

*CURSYM 

Indicates that a floating currency symbol is to be used. The actual symbol will be the one specified on the control specification in the CURSYM keyword, or the default, '$'. When *CURSYM is specified, the currency symbol is placed in the the result just before the first significant digit. For example, 
| %EDITC(0012.5 : 'K' : *CURSYM) returns ' $12.5 ' . 

currency-symbol 

Indicates that floating currency is to be used with the provided currency symbol. It must be a 1-byte character constant (literal, named constant or expression that can be evaluated at compile time). For example, %EDITC(0012.5 : 'K' : 'X') returns ' X12.5 '. 


| The result of %EDITC is always the same length, and may contain leading 
| and trailing blanks. For example, %EDITC(NUM : 'A' : '$') might return 
| '$1,234.56CR' for one value of NUM and ' $4.56 ' for another value. 

Float expressions are not allowed in the first parameter (you can use %DEC to convert a float to an editable format). In the second parameter, the edit code is specified as a character constant; supported edit codes are: 'A' - 'D', 'J' - 'Q', 'X' - 'Z', '1' - '9'. The constant can be a literal, named constant or an expression whose value can be determined at compile time. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

 | |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

 | |   D msg             S            100A                                                            | 

 | |   D salary          S              9P 2 INZ(1000)                                                | 

 | |    * If the value of salary is 1000, then the value of salary * 12                               | 

 | |    * is 12000.00. The edited version of salary * 12 using the A edit                             | 

 | |    * code with floating currency is ' $12,000.00 '.                                              | 

 | |    * The value of msg is 'The annual salary is $12,000.00'                                       | 

 | |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

 | |   C                   EVAL      msg = 'The annual salary is '                                    | 

 | |   C                                + %trim(%editc(salary * 12                                    | 

 | |   C                                              :'A': *CURSYM))                                 | 

   |                                                                                                  | 

 | |    * In the next example, the value of msg is 'The annual salary is &12,000.00'                  | 

 | |   C                   EVAL      msg = 'The annual salary is '                                    | 

 | |   C                                + %trim(%editc(salary * 12                                    | 

 | |   C                                              :'A': '&'))                                     | 

   |                                                                                                  | 

 | |    * In the next example, the value of msg is 'Salary is $*****12,000.00'                        | 

 | |    * Note that the '$' comes from the text, not from the edit code.                              | 

 | |   C                   EVAL      msg = 'Salary is $'                                              | 

 | |   C                                + %trim(%editc(salary * 12                                    | 

 | |   C                                               :'B': *ASTFILL))                               | 

   |                                                                                                  | 

 | |    * In the next example, the value of msg is 'The date is 1/14/1999'                            | 

 | |   C                   EVAL      msg = 'The date is '                                             | 

 | |   C                                + %trim(%editc(*date : 'Y'))                                  | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

 | Figure 138. %EDITC Example 1 

A common requirement is to edit a field as follows: 

· Leading zeros are suppressed 

· Parentheses are placed around the value if it is negative 

The following accomplishes this using an %EDITC in a subprocedure: 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D neg             S              5P 2      inz(-12.3)                                          | 

   |   D pos             S              5P 2      inz(54.32)                                          | 

   |   D editparens      PR            50A                                                            | 

   |   D    val                        30P 2      value                                               | 

   |   D editedVal       S             10A                                                            | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   EVAL      editedVal = editparens(neg)                                      | 

   |    * Now editedVal has the value '(12.30)   '                                                    | 

   |   C                   EVAL      editedVal = editparens(pos)                                      | 

   |    * Now editedVal has the value ' 54.32    '                                                    | 

   |    *---------------------------------------------------------------                              | 

   |    * Subprocedure EDITPARENS                                                                     | 

   |    *---------------------------------------------------------------                              | 

   |   P editparens      B                                                                            | 

   |   D editparens      PI            50A                                                            | 

   |   D    val                        30P 2      value                                               | 

   |   D lparen          S              1A        inz(' ')                                            | 

   |   D rparen          S              1A        inz(' ')                                            | 

   |   D res             S             50A                                                            | 

   |    * Use parentheses if the value is negative                                                    | 

   |   C                   IF        val < 0                                                          | 

   |   C                   EVAL      lparen = '('                                                     | 

   |   C                   EVAL      rparen = ')'                                                     | 

   |   C                   ENDIF                                                                      | 

   |                                                                                                  | 

   |    * Return the edited value                                                                     | 

   |    * Note that the '1' edit code does not include a sign so we                                   | 

   |    * don't have to calculate the absolute value.                                                 | 

   |   C                   RETURN    lparen             +                                             | 

   |   C                             %editc(val : '1')  +                                             | 

   |   C                             rparen                                                           | 

   |   P editparens      E                                                                            | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 139. %EDITC Example 2 


8 %EDITFLT (Convert to Float External Representation)

     %EDITFLT(numeric expression) 

%EDITFLT converts the value of the numeric expression to the character external display representation of float. The result is either 14 or 23 characters. If the argument is a 4-byte float field, the result is 14 characters. Otherwise, it is 23 characters. 

If specified as a parameter to a definition specification keyword, the parameter must be a numeric literal, float literal, or numeric valued constant name or built-in function. When specified in an expression, constant folding is applied if the numeric expression has a constant value. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D f8              s              8f   inz (50000)                                              | 

   |   D string          s             40a                                                            | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      string = 'Float value is '                                       | 

   |   C                                    + %editflt (f8 - 4e4) + '.'                               | 

   |    * Value of "string" is 'Float value is +1.000000000000000E+004. '                             | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 140. %EDITFLT Example 


9 %EDITW (Edit Value Using an Editword)

     %EDITW(numeric : editword) 

This function returns a character result representing the numeric value edited according to the edit word. The rules for the numeric value and edit word are identical to those for editing numeric values in output specifications. 

Float expressions are not allowed in the first parameter. Use %DEC to convert a float to an editable format. 

The edit word must be a character constant. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D amount          S             30A                                                            | 

   |   D salary          S              9P 2                                                          | 

   |   D editwd          C                    '$ ,   ,  **Dollars&  &Cents'                           | 

   |    * If the value of salary is 2451.53, then the edited version of                               | 

   |    * (salary * 12) is '$***29,418*Dollars 36 Cents'. The value of                                | 

   |    * amount is 'The annual salary is $***29,418*Dollars 36 Cents'.                               | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   EVAL      amount = 'The annual salary is '                                 | 

   |   C                              + %editw(salary * 12 : editwd)                                  | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 141. %EDITW Example 


10 %ELEM (Get Number of Elements)

     %ELEM(table_name) 

     %ELEM(array_name) 

     %ELEM(multiple_occurrence_data_structure_name) 

%ELEM returns the number of elements in the specified array, table, or multiple-occurrence data structure. The value returned is in unsigned integer format (type U). It may be specified anywhere a numeric constant is allowed in the definition specification or in an expression in the extended factor 2 field. 

The parameter must be the name of an array, table, or multiple occurrence data structure. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords++++++++++++++++++++++++++                       | 

   |   D arr1d           S             20    DIM(10)                                                  | 

   |   D table           S             10    DIM(20) ctdata                                           | 

   |   D mds             DS            20    occurs(30)                                               | 

   |   D num             S              5P                                                            | 

   |    * like_array will be defined with a dimension of 10.                                          | 

   |    * array_dims will be defined with a value of 10.                                              | 

   |   D like_array      S                   like(arr1d) dim(%elem(arr1d))                            | 

   |   D array_dims      C                   const (%elem (arr1d))                                    | 

   |   C*L0N01Factor1+++++++Opcode(E)+Extended-factor2++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * In the following examples num will be equal to 10, 20, and 30.                              | 

   |    *                                                                                             | 

   |   C                   EVAL      num = %elem (arr1d)                                              | 

   |   C                   EVAL      num = %elem (table)                                              | 

   |   C                   EVAL      num = %elem (mds)                                                | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 142. %ELEM Example 


11 %EOF (Return End or Beginning of File Condition)

     %EOF{(file_name)} 

%EOF returns '1' if the most recent read operation or write to a subfile ended in an end of file or beginning of file condition; otherwise, it returns '0'. 

The operations that set %EOF are: 

· "READ (Read a Record)"  

· "READC (Read Next Changed Record)"  

· "READE (Read Equal Key)" 
· "READP (Read Prior Record)" 

· "READPE (Read Prior Equal)" 

· "WRITE (Create New Records)" (subfile only). 

When a full-procedural file is specified, this function returns '1' if the previous operation in the list above, for the specified file, resulted in an end of file or beginning of file condition. For primary and secondary files, %EOF is available only if the file name is specified. It is set to '1' if the most recent input operation during *GETIN processing resulted in an end of file or beginning of file condition. Otherwise, it returns '0'. 

This function is allowed for input, update, and record-address files; and for display files allowing WRITE to subfile records. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   FFilename++IPEASFRlen+LKlen+AIDevice+.Keywords++++++++++++++++++++++++++++                     | 

   |    * File INFILE has record format INREC                                                         | 

   |   FINFILE    IF   E             DISK                                                             | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * Read a record                                                                               | 

   |   C                   READ      INREC                                                            | 

   |    * If end-of-file was reached ...                                                              | 

   |   C                   IF        %EOF                                                             | 

   |   C ...                                                                                          | 

   |   C                   ENDIF                                                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 143. %EOF without a Filename Parameter 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   FFilename++IPEASFRlen+LKlen+AIDevice+.Keywords++++++++++++++++++++++++++++                     | 

   |    * This program is comparing two files                                                         | 

   |    *                                                                                             | 

   |   FFILE1     IF   E             DISK                                                             | 

   |   FFILE2     IF   E             DISK                                                             | 

   |   F                                                                                              | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * Loop until either FILE1 or FILE2 has reached end-of-file                                    | 

   |   C                   DOU       %EOF(FILE1) OR %EOF(FILE2)                                       | 

   |    * Read a record from each file and compare the records                                        | 

   |    *                                                                                             | 

   |   C                   READ      REC1                                                             | 

   |   C                   READ      REC2                                                             | 

   |   C                   SELECT                                                                     | 

   |   C                   WHEN      %EOF(FILE1) AND %EOF(FILE2)                                      | 

   |    * Both files have reached end-of-file                                                         | 

   |   C                   EXSR      EndCompare                                                       | 

   |   C                   WHEN      %EOF(FILE1)                                                      | 

   |    * FILE1 is shorter than FILE2                                                                 | 

   |   C                   EXSR      F1Short                                                          | 

   |   C                   WHEN      %EOF(FILE2)                                                      | 

   |    * FILE2 is shorter than FILE1                                                                 | 

   |   C                   EXSR      F2Short                                                          | 

   |   C                   OTHER                                                                      | 

   |    * Both files still have records to be compared                                                | 

   |   C                   EXSR      CompareRecs                                                      | 

   |   C                   ENDSL                                                                      | 

   |   C                   ENDDO                                                                      | 

   |    ...                                                                                           | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 144. %EOF with a Filename Parameter 


12 %EQUAL (Return Exact Match Condition)

     %EQUAL{(file_name)} 

%EQUAL returns '1' if the most recent relevant operation found an exact match; otherwise, it returns '0'. 

The operations that set %EQUAL are: 

· "SETLL (Set Lower Limit)"  

· "LOOKUP (Look Up a Table or Array Element)" 

If %EQUAL is used without the optional file_name parameter, then it returns the value set for the most recent relevant operation. 

For the SETLL operation, this function returns '1' if a record is present whose key or relative record number is equal to the search argument. 

For the LOOKUP operation with the EQ indicator specified, this function returns '1' if an element is found that exactly matches the search argument. 

If a file name is specified, this function applies to the most recent SETLL operation for the specified file. This function is allowed only for files that allow the SETLL operation code. 

__________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   FFilename++IPEASFRlen+LKlen+AIDevice+.Keywords++++++++++++++++++++++++++++                     | 

   |    * File CUSTS has record format CUSTREC                                                        | 

   |   FCUSTS     IF   E           K DISK                                                             | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * Check if the file contains a record with a key matching Cust                                | 

   |   C     Cust          SETLL     CUSTREC                                                          | 

   |   C                   IF        %EQUAL                                                           | 

   |   C ... an exact match was found in the file                                                     | 

   |   C                   ENDIF                                                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 145. %EQUAL with SETLL Example 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |        DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++++               | 

   |        D TabNames        S             10A   DIM(5) CTDATA ASCEND                                | 

   |        D SearchName      S             10A                                                       | 

   |         * Position the table at or near SearchName                                               | 

   |         * Here are the results of this program for different values                              | 

   |         * of SearchName:                                                                         | 

   |         *   SearchName   |   DSPLY                                                               | 

   |         *   -------------+-------------------------------                                        | 

   |         *   'Catherine ' |   'Next greater     Martha'                                           | 

   |         *   'Andrea    ' |   'Exact            Andrea'                                           | 

   |         *   'Thomas    ' |   'Not found        Thomas'                                           | 

   |        CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....               | 

   |        C     SearchName    LOOKUP    TabNames                           10  10                   | 

   |        C                   SELECT                                                                | 

   |        C                   WHEN      %EQUAL                                                      | 

   |         * An exact match was found                                                               | 

   |        C     'Exact       'DSPLY                   TabNames                                      | 

   |        C                   WHEN      %FOUND                                                      | 

   |         * A name was found greater than SearchName                                               | 

   |        C     'Next greater'DSPLY                   TabNames                                      | 

   |        C                   OTHER                                                                 | 

   |         * Not found.  SearchName is greater than all the names in the table                      | 

   |        C     'Not found   'DSPLY                   SearchName                                    | 

   |        C                   ENDSL                                                                 | 

   |        C                   RETURN                                                                | 

   |   **CTDATA TabNames                                                                              | 

   |   Alexander                                                                                      | 

   |   Andrea                                                                                         | 

   |   Bohdan                                                                                         | 

   |   Martha                                                                                         | 

   |   Samuel                                                                                         | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 146. %EQUAL and %FOUND with LOOKUP Example 


13 %ERROR (Return Error Condition)

%ERROR returns '1' if the most recent operation with extender 'E' specified resulted in an error condition. This is the same as the error indicator being set on for the operation. Before an operation with extender 'E' specified begins, %ERROR is set to return '0' and remains unchanged following the operation if no error occurs. All operations that allow an error indicator can also set the %ERROR built-in function. The CALLP operation can also set %ERROR. 

For examples of the %ERROR built-in function, see Figure 163 in topic 27 and Figure 164 in topic 27. 


14 %FLOAT (Convert to Floating Format)

     %FLOAT(numeric expression) 

%FLOAT converts the value of the numeric expression to float format. This built-in function may only be used in expressions. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D p1              s             15p 0 inz (1)                                                  | 

   |   D p2              s             25p13 inz (3)                                                  | 

   |   D result1         s             15p 5                                                          | 

   |   D result2         s             15p 5                                                          | 

   |   D result3         s             15p 5                                                          | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      result1 = p1 / p2                                                | 

   |   C                   eval      result2 = %float (p1) / p2                                       | 

   |   C                   eval      result3 = %float (p1 / p2)                                       | 

   |    * The value of "result1" is now  0.33000.                                                     | 

   |    * The value of "result2" is now  0.33333.                                                     | 

   |    * The value of "result3" is now  0.33333.                                                     | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 147. %FLOAT Example 


15 %FOUND (Return Found Condition)

     %FOUND{(file_name)} 

%FOUND returns '1' if the most recent relevant file operation found a record, a string operation found a match, or a search operation found an element. Otherwise, this function returns '0'. 

The operations that set %FOUND are: 

· File operations: 

· "CHAIN (Random Retrieval from a File)" 
· "DELETE (Delete Record)" 

· "SETGT (Set Greater Than)" 

· "SETLL (Set Lower Limit)"  

· String operations: 

· "CHECK (Check Characters)"  

· "CHECKR (Check Reverse)"  

· "SCAN (Scan String)" 

Note: Built-in function %SCAN does not change the value of %FOUND. 

· Search operations: 

· "LOOKUP (Look Up a Table or Array Element)"  

If %FOUND is used without the optional file_name parameter, then it returns the value set for the most recent relevant operation. When a file_name is specified, then it applies to the most recent relevant operation on that file. 

For file operations, %FOUND is opposite in function to the "no record found NR" indicator. 

For string operations, %FOUND is the same in function as the "found FD" indicator. 

For the LOOKUP operation, %FOUND returns '1' if the operation found an element satisfying the search conditions. For an example of %FOUND with LOOKUP, see Figure 146. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   FFilename++IPEASFRlen+LKlen+AIDevice+.Keywords++++++++++++++++++++++++++++                     | 

   |    * File CUSTS has record format CUSTREC                                                        | 

   |   FCUSTS     IF   E           K DISK                                                             | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * Check if the customer is in the file                                                        | 

   |   C     Cust          CHAIN     CUSTREC                                                          | 

   |   C                   IF        %FOUND                                                           | 

   |   C ...                                                                                          | 

   |   C                   ENDIF                                                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 148. %FOUND used to Test a File Operation without a Parameter 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   FFilename++IPEASFRlen+LKlen+AIDevice+.Keywords++++++++++++++++++++++++++++                     | 

   |    * File MASTER has all the customers                                                           | 

   |    * File GOLD has only the "privileged" customers                                               | 

   |   FMASTER    IF   E           K DISK                                                             | 

   |   FGOLD      IF   E           K DISK                                                             | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * Check if the customer exists, but is not a privileged customer                              | 

   |   C     Cust          CHAIN     MASTREC                                                          | 

   |   C     Cust          CHAIN     GOLDREC                                                          | 

   |   * Note that the file name is used for %FOUND, not the record name                              | 

   |   C                   IF        %FOUND(MASTER) AND NOT %FOUND(GOLD)                              | 

   |   C ...                                                                                          | 

   |   C                   ENDIF                                                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 149. %FOUND used to Test a File Operation with a Parameter 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D Numbers         C                   '0123456789'                                             | 

   |   D Position        S              5I 0                                                          | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * If the actual position of the name is not required, just use                                | 

   |    * %FOUND to test the results of the SCAN operation.                                           | 

   |    * If Name has the value 'Barbara' and Line has the value                                      | 

   |    * 'in the city of Toronto.      ', then %FOUND will return '0'.                               | 

   |    * If Line has the value 'the city of Toronto where Barbara lives, '                           | 

   |    * then %FOUND will return '1'.                                                                | 

   |   C     Name          SCAN      Line                                                             | 

   |   C                   IF        %FOUND                                                           | 

   |   C                   EXSR      PutLine                                                          | 

   |   C                   ENDIF                                                                      | 

   |    * If Value contains the value '12345.67', Position would be set                               | 

   |    * to 6 and %FOUND would return the value '1'.                                                 | 

   |    * If Value contains the value '10203040', Position would be set                               | 

   |    * to 0 and %FOUND would return the value '0'.                                                 | 

   |   C     Numbers       CHECK     Value         Position                                           | 

   |   C                   IF        %FOUND                                                           | 

   |   C                   EXSR      HandleNonNum                                                     | 

   |   C                   ENDIF                                                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 150. %FOUND used to Test a String Operation 


16 %GRAPH (Convert to Graphic Value)

 |   %GRAPH(char-expr | graph-expr | UCS-2-expr { : ccsid }) 


| %GRAPH converts the value of the expression from character, graphic, or 
| UCS-2 and returns a graphic value. The result is varying length if the 
| parameter is varying length. 


| The second parameter, ccsid, is optional and indicates the CCSID of the 
| resulting expression. The CCSID defaults to the graphic CCSID related to 
| the CCSID of the job. If CCSID(*GRAPH : *IGNORE) is specified on the 
| control specification or assumed for the module, the %GRAPH built-in is 
| not allowed. 


| If the parameter is a constant, the conversion will be done at compile 
| time. In this case, the CCSID is the graphic CCSID related to the CCSID of 
| the source file. 


| If the conversion results in substitution characters, a warning message is 
| issued at compile time. At run time, status 00050 is set and no error 
| message is issued. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

 | |   HKeywords++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++                        | 

 | |   H CCSID(*GRAPH : 300)                                                                          | 

 | |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++                        | 

 | |   D  char           S              5A   INZ('abcde')                                             | 

 | |    * The %GRAPH built-in function is used to initialize a graphic field                          | 

 | |   D  graph          S             10G   INZ(%GRAPH('oAABBCCDDEEi'))                              | 

 | |   D  ufield         S              2C   INZ(%UCS2('oFFGGi'))                                     | 

 | |   D  graph2         S              2G   CCSID(4396) INZ(*HIVAL)                                  | 

 | |   D  isEqual        S              1N                                                            | 

 | |   D  proc           PR                                                                           | 

 | |   D    gparm                       2G   CCSID(4396) VALUE                                        | 

 | |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

 | |   C                   EVAL      graph = %GRAPH(char) + %GRAPH(ufield)                            | 

 | |    * graph now has 7 graphic characters AABBCCDDEEFFGG.                                          | 

   |                                                                                                  | 

 | |   C                   EVAL      isEqual = graph = %GRAPH(graph2 : 300)                           | 

 | |    * The result of the %GRAPH built-in function is the value of                                  | 

 | |    * graph2, converted from CCSID 4396 to CCSID 300.                                             | 

   |                                                                                                  | 

 | |   C                   EVAL      graph2 = graph                                                   | 

 | |    * The value of graph is converted from CCSID 300 to CCSID 4396                                | 

 | |    * and stored in graph2.                                                                       | 

 | |    * This conversion is performed implicitly by the compiler.                                    | 

   |                                                                                                  | 

 | |   C                   CALLP     proc(graph)                                                      | 

 | |    * The value of graph is converted from CCSID 300 to CCSID 4396                                | 

 | |    * implicitly, as part of passing the parameter by value.                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

 | Figure 151. %GRAPH Examples 


17 %INT (Convert to Integer Format)

     %INT(numeric expression) 

%INT converts the value of the numeric expression to integer. Any decimal digits are truncated. This built-in function may only be used in expressions. %INT can be used to truncate the decimal positions from a float or decimal value allowing it to be used as an array index. 

Subtopics: 

· 17.1 %INTH (Convert to Integer Format with Half Adjust) 

17.1 %INTH (Convert to Integer Format with Half Adjust)

     %INTH(numeric expression) 

%INTH is the same as %INT except that if the numeric expression is a decimal or float value, half adjust is applied to the value of the numeric expression when converting to integer type. No message is issued if half adjust cannot be performed. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D p7              s              7p 3 inz (1234.567)                                           | 

   |   D s9              s              9s 5 inz (73.73442)                                           | 

   |   D f8              s              8f   inz (123.789)                                            | 

   |   D result1         s             15p 5                                                          | 

   |   D result2         s             15p 5                                                          | 

   |   D result3         s             15p 5                                                          | 

   |   D array           s              1a   dim (200)                                                | 

   |   D a               s              1a                                                            | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      result1 = %int (p7) + 0.011                                      | 

   |   C                   eval      result2 = %int (s9)                                              | 

   |   C                   eval      result3 = %inth (f8)                                             | 

   |    * The value of "result1" is now 1234.01100.                                                   | 

   |    * The value of "result2" is now   73.00000                                                    | 

   |    * The value of "result3" is now  124.00000.                                                   | 

   |   C                   eval      a = array (%inth (f8))                                           | 

   |    * %INT and %INTH can be used as array indexes                                                 | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 152. %INT and %INTH Example 

18 %LEN (Get or Set Length)

     %LEN(expression) 

%LEN can be used to get the length of a variable expression or to set the current length of a variable-length field. 

The parameter must not be a figurative constant. 

Subtopics: 

· 18.1 %LEN Used for its Value 

· 18.2 %LEN Used to Set the Length of Variable-Length Fields 

18.1 %LEN Used for its Value

When used on the right-hand side of an expression, this function returns the number of digits or characters of the variable expression. 

For numeric expressions, the value returned represents the precision of the expression and not necessarily the actual number of significant digits. For a float variable or expression, the value returned is either 4 or 8. When the parameter is a numeric literal, the length returned is the number of digits of the literal. 


| For character, graphic, or UCS-2 expressions the value returned is the number of characters in the value of the expression. For variable-length values, such as the value returned from a built-in function or a 
| variable-length field, the value returned by %LEN is the current length of 
| the character, graphic, or UCS-2 value. 

Note that if the parameter is a built-in function or expression that has a value computable at compile-time, the length returned is the actual number of digits of the constant value rather than the maximum possible value that could be returned by the expression. 

For all other data types, the value returned is the number of bytes of the value. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D num1            S              7P 2                                                          | 

   |   D num2            S              5S 1                                                          | 

   |   D num3            S              5I 0 inz(2)                                                   | 

   |   D chr1            S             10A   inz('Toronto   ')                                        | 

   |   D chr2            S             10A   inz('Munich    ')                                        | 

   |   D ptr             S               *                                                            | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    * Numeric expressions:                                                                        | 

   |   C                   eval      num1 = %len(num1)           <===  7                              | 

   |   C                   eval      num1 = %decpos(num2)        <===  1                              | 

   |   C                   eval      num1 = %len(num1*num2)      <=== 12                              | 

   |   C                   eval      num1 = %decpos(num1*num2)   <===  3                              | 

   |    * Character expressions:                                                                      | 

   |   C                   eval      num1 = %len(chr1)           <=== 10                              | 

   |   C                   eval      num1 = %len(chr1+chr2)      <=== 20                              | 

   |   C                   eval      num1 = %len(%trim(chr1))    <===  7                              | 

   |   C                   eval      num1 = %len(%subst(chr1:1:num3)                                  | 

   |   C                               + ' ' + %trim(chr2))      <===  9                              | 

   |                                                                                                  | 

   |    * %len and %decpos can be useful with other built-in functions:                               | 

   |                                                                                                  | 

   |    * Although this division is performed in float, the result is                                 | 

   |    * converted to the same precision as the result of the eval:                                  | 

   |   C                   eval      num1 = 27 + %dec (%float(num1)/num3                              | 

   |   C                                             : %len(num1)                                     | 

   |   C                                             : %decpos(num1))                                 | 

   |                                                                                                  | 

   |    * Allocate sufficient space to hold the result of the catenation                              | 

   |    * (plus an extra byte for a trailing null character):                                         | 

   |   C                   eval      num3 = %len(chr1+chr2)+1                                         | 

   |   C                   alloc     num3          ptr                                                | 

   |   C                   eval      %str(ptr : num3) = chr1 + chr2                                   | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 153. %DECPOS and %LEN Example 


18.2 %LEN Used to Set the Length of Variable-Length Fields

When used on the left-hand side of an expression, this function sets the current length of a variable-length field. If the set length is greater than the current length, the characters in the field between the old length and the new length are set to blanks. 

Note: %LEN can only be used on the left-hand-side of an expression when the parameter is variable length. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |   D city            S             40A   VARYING INZ('North York')                                | 

   |   D n1              S              5i 0                                                          | 

   |                                                                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    * %LEN used to get the current length of a variable-length field:                             | 

   |   C                   EVAL      n1 = %LEN(city)                                                  | 

   |    * Current length, n1 = 10                                                                     | 

   |    *                                                                                             | 

   |    * %LEN used to set the current length of a variable-length field:                             | 

   |   C                   EVAL      %LEN(city) = 5                                                   | 

   |    * city = 'North' (length is 5)                                                                | 

   |    *                                                                                             | 

   |   C                   EVAL      %LEN(city) = 15                                                  | 

   |    * city = 'North          '  (length is 15)                                                    | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 154. %LEN with Variable-Length Field Example 

19 %NULLIND (Query or Set Null Indicator)

     %NULLIND(fieldname) 

The %NULLIND built-in function can be used to query or set the null indicator for null-capable fields. This built-in function can only be used if the ALWNULL(*USRCTL) keyword is specified on a control specification or as a command parameter. The fieldname can be a null-capable array element, data structure, stand-alone field, subfield, or multiple occurrence data structure. 

%NULLIND can only be used in expressions in extended factor 2. 

When used on the right-hand side of an expression, this function returns the setting of the null indicator for the null-capable field. The setting can be *ON or *OFF. 

When used on the left-hand side of an expression, this function can be used to set the null indicator for null-capable fields to *ON or *OFF. The content of a null-capable field remains unchanged. 

See "Database Null Value Support"  for more information on handling records with null-capable fields and keys. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * Test the null indicator for a null-capable field.                                           | 

   |    *                                                                                             | 

   |   C                   IF        %NULLIND(fieldname1)                                             | 

   |   C                   :                                                                          | 

   |   C                   ENDIF                                                                      | 

   |    *                                                                                             | 

   |    * Set the null indicator for a null-capable field.                                            | 

   |    *                                                                                             | 

   |   C                   EVAL      %NULLIND(fieldname1) = *ON                                       | 

   |   C                   EVAL      %NULLIND(fieldname2) = *OFF                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 155. %NULLIND Example 


20 %OPEN (Return File Open Condition)

     %OPEN(file_name) 

%OPEN returns '1' if the specified file is open. A file is considered "open" if it has been opened by the RPG program during initialization or by an OPEN operation, and has not subsequently been closed. If the file is conditioned by an external indicator and the external indicator was off at program initialization, the file is considered closed, and %OPEN returns '0'. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   FFilename++IPEASFRlen+LKlen+AIDevice+.Keywords++++++++++++++++++++++++++++                     | 

   |    * The printer file is opened in the calculation specifications                                | 

   |   FQSYSPRT   O    F  132        PRINTER USROPN                                                   | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * Open the file if it is not already open                                                     | 

   |   C                   IF        NOT %OPEN(QSYSPRT)                                               | 

   |   C                   OPEN      QSYSPRT                                                          | 

   |   C                   ENDIF                                                                      | 

   |    ...                                                                                           | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 156. %OPEN Example 

21 %PADDR (Get Procedure Address)

     %PADDR(string) 

%PADDR returns a value of type procedure pointer. The value is the address of the entry point specified as the argument. 

%PADDR may be compared with and assigned to only items of type procedure pointer. 

The parameter to %PADDR must be a character or hexadecimal literal or a constant name that represents a character or hexadecimal literal. The entry point name specified by the character string must be found at program bind time and must be in the correct case. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D                                                                                              | 

   |   D PROC            S               *   PROCPTR                                                  | 

   |   D                                     INZ (%PADDR ('FIRSTPROG'))                               | 

   |   D PROC1           S               *   PROCPTR                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    *  The following statement calls procedure 'FIRSTPROG'.                                       | 

   |    *                                                                                             | 

   |   C                   CALLB     PROC                                                             | 

   |    *-----------------------------------------------------------------                            | 

   |    * The following statements call procedure 'NextProg'.                                         | 

   |    * This a C procedure and is in mixed case.  Note that                                         | 

   |    * the procedure name is case sensitive.                                                       | 

   |    *                                                                                             | 

   |   C                   EVAL      PROC1 = %PADDR ('NextProg')                                      | 

   |   C                   CALLB     PROC1                                                            | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 157. %PADDR Example 


22 %PARMS (Return Number of Parameters)

%PARMS returns the number of parameters that were passed to the procedure in which %PARMS is used. For the main procedure, %PARMS is the same as *PARMS. 

The value returned by %PARMS is not available if the program or procedure that calls %PARMS does not pass a minimal operational descriptor. The ILE RPGcompiler always passes one, but other languages do not. So if the caller is written in another ILE language, it will need to pass an operational descriptor on the call. If the operational descriptor is not passed, the value returned by %PARMS cannot be trusted. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |    * Prototype for procedure MaxInt which calculates the maximum                                 | 

   |    * value of its parameters (at least 2 parameters must be passed)                              | 

   |   D MaxInt          PR            10I 0                                                          | 

   |   D  p1                           10I 0 VALUE                                                    | 

   |   D  p2                           10I 0 VALUE                                                    | 

   |   D  p3                           10I 0 VALUE OPTIONS(*NOPASS)                                   | 

   |   D  p4                           10I 0 VALUE OPTIONS(*NOPASS)                                   | 

   |   D  p5                           10I 0 VALUE OPTIONS(*NOPASS)                                   | 

   |   D Fld1            S             10A   DIM(40)                                                  | 

   |   D Fld2            S             20A                                                            | 

   |   D Fld3            S            100A                                                            | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |   C     *ENTRY        PLIST                                                                      | 

   |   C                   PARM                    MaxSize          10 0                              | 

   |    * Make sure the main procedure was passed a parameter                                         | 

   |   C                   IF        %PARMS < 1                                                       | 

   |   C     'No parms'    DSPLY                                                                      | 

   |   C                   RETURN                                                                     | 

   |   C                   ENDIF                                                                      | 

   |    * Determine the maximum size of Fld1, Fld2 and Fld3                                           | 

   |   C                   EVAL      MaxSize = MaxInt(%size(Fld1:*ALL) :                              | 

   |   C                                             %size(Fld2) :                                    | 

   |   C                                             %size(Fld3))                                     | 

   |   C     'MaxSize is'  DSPLY                   MaxSize                                            | 

   |   C                   RETURN                                                                     | 

   |                                                                                                  | 

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |    *----------------------------------------------------------------                             | 

   |    * MaxInt - return the maximum value of the passed parameters                                  | 

   |    *----------------------------------------------------------------                             | 

   |   P MaxInt          B                                                                            | 

   |   D MaxInt          PI            10I 0                                                          | 

   |   D  p1                           10I 0 VALUE                                                    | 

   |   D  p2                           10I 0 VALUE                                                    | 

   |   D  p3                           10I 0 VALUE OPTIONS(*NOPASS)                                   | 

   |   D  p4                           10I 0 VALUE OPTIONS(*NOPASS)                                   | 

   |   D  p5                           10I 0 VALUE OPTIONS(*NOPASS)                                   | 

   |   D Max             S             10I 0 INZ(*LOVAL)                                              | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    * Branch to the point in the calculations where we will never                                 | 

   |    * access unpassed parameters.                                                                 | 

   |   C                   SELECT                                                                     | 

   |   C                   WHEN      %PARMS = 2                                                       | 

   |   C                   GOTO      PARMS2                                                           | 

   |   C                   WHEN      %PARMS = 3                                                       | 

   |   C                   GOTO      PARMS3                                                           | 

   |   C                   WHEN      %PARMS = 4                                                       | 

   |   C                   GOTO      PARMS4                                                           | 

   |   C                   WHEN      %PARMS = 5                                                       | 

   |   C                   GOTO      PARMS5                                                           | 

   |   C                   ENDSL                                                                      | 

   |    * Determine the maximum value.  Max was initialized to *LOVAL.                                | 

   |   C     PARMS5        TAG                                                                        | 

   |   C                   IF        p5 > Max                                                         | 

   |   C                   EVAL      Max = p5                                                         | 

   |   C                   ENDIF                                                                      | 

   |    *                                                                                             | 

   |   C     PARMS4        TAG                                                                        | 

   |   C                   IF        p4 > Max                                                         | 

   |   C                   EVAL      Max = p4                                                         | 

   |   C                   ENDIF                                                                      | 

   |    *                                                                                             | 

   |   C     PARMS3        TAG                                                                        | 

   |   C                   IF        p3 > Max                                                         | 

   |   C                   EVAL      Max = p3                                                         | 

   |   C                   ENDIF                                                                      | 

   |    *                                                                                             | 

   |   C     PARMS2        TAG                                                                        | 

   |   C                   IF        p2 > Max                                                         | 

   |   C                   EVAL      Max = p2                                                         | 

   |   C                   ENDIF                                                                      | 

   |   C                   IF        p1 > Max                                                         | 

   |   C                   EVAL      Max = p1                                                         | 

   |   C                   ENDIF                                                                      | 

   |   C                   RETURN    Max                                                              | 

   |   P MaxInt          E                                                                            | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 158. %PARMS Example 


23 %REM (Return Integer Remainder)

 |   %REM(n:m) 


| %REM returns the remainder that results from dividing operands n by m. The 
| two operands must be numeric values with zero decimal positions. If either 
| operand is a packed, zoned, or binary numeric value, the result is packed 
| numeric. If either operand is an integer numeric value, the result is 
| integer. Otherwise, the result is unsigned numeric. Float numeric operands 
| are not allowed. The result has the same sign as the dividend. (See also 
| "%DIV (Return Integer Portion of Quotient)" in topic 6.) 


| %REM and %DIV have the following relationship: 

 |   %REM(A:B) = A - (%DIV(A:B) * B) 


| If the operands are constants that can fit in 8-byte integer or unsigned 
| fields, constant folding is applied to the built-in function. In this 
| case, the %REM built-in function can be coded in the definition 
| specifications. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

 | |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

 | |    *                                                                                             | 

 | |   D A               S             10I 0 INZ(123)                                                 | 

 | |   D B               S             10I 0 INZ(27)                                                  | 

 | |   D DIV             S             10I 0                                                          | 

 | |   D REM             S             10I 0                                                          | 

 | |   D E               S             10I 0                                                          | 

 | |    *                                                                                             | 

 | |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

 | |    *                                                                                             | 

 | |   C                   EVAL      DIV = %DIV(A:B)                                                  | 

 | |   C                   EVAL      REM = %REM(A:B)                                                  | 

 | |   C                   EVAL      E = DIV*B + REM                                                  | 

 | |    * Now, DIV = 4, REM = 15, E = 123                                                             | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

 | Figure 159. %DIV and %REM Example 

24 %REPLACE (Replace Character String)

     %REPLACE(replacement string: source string{:start position {:source 

     length to replace}}) 

%REPLACE returns the character string produced by inserting a replacement string into the source string, starting at the start position and replacing the specified number of characters. 


| The first and second parameter must be of type character, graphic, or 
| UCS-2 and can be in either fixed- or variable-length format. The second parameter must be the same type as the first. 

The third parameter represents the starting position, measured in characters, for the replacement string. If it is not specified, the starting position is at the beginning of the source string. The value may range from one to the current length of the source string plus one. 

The fourth parameter represents the number of characters in the source string to be replaced. If zero is specified, then the replacement string is inserted before the specified starting position. If the parameter is not specified, the number of characters replaced is the same as the length of the replacement string. The value must be greater than or equal to zero, and less than or equal to the current length of the source string. 

The starting position and length may be any numeric value or numeric expression with no decimal positions. 

The returned value is varying length if the source string or replacement string are varying length, or if the start position or source length to replace are variables. Otherwise, the result is fixed length. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D var1            S             30A   INZ('Windsor') VARYING                                   | 

   |   D var2            S             30A   INZ('Ontario') VARYING                                   | 

   |   D var3            S             30A   INZ('Canada') VARYING                                    | 

   |   D fixed1          S             15A   INZ('California')                                        | 

   |   D date            S               D   INZ(D'1997-02-03')                                       | 

   |   D result          S            100A   VARYING                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |   C                                                                                              | 

   |   C                   EVAL      result = var1 + ', ' + 'ON'                                      | 

   |    * result = 'Windsor, ON'                                                                      | 

   |    *                                                                                             | 

   |    * %REPLACE with 2 parameters to replace text at begining of string:                           | 

   |   C                   EVAL      result = %REPLACE('Toronto': result)                             | 

   |    * result = 'Toronto, ON'                                                                      | 

   |    *                                                                                             | 

   |    * %REPLACE with 3 parameters to replace text at specified position:                           | 

   |   C                   EVAL      result = %REPLACE(var3: result:                                  | 

   |   C                                            %SCAN(',': result)+2)                             | 

   |    * result = 'Toronto, Canada'                                                                  | 

   |    *                                                                                             | 

   |    * %REPLACE with 4 parameters to insert text:                                                  | 

   |   C                   EVAL      result = %REPLACE(', '+var2: result:                             | 

   |   C                                            %SCAN(',': result): 0)                            | 

   |    * result = 'Toronto, Ontario, Canada'                                                         | 

   |    *                                                                                             | 

   |    * %REPLACE with 4 parameters to replace strings with different lengths:                       | 

   |   C                   EVAL      result = %REPLACE('Scarborough': result:                         | 

   |   C                                            1: %SCAN(',': result)-1)                          | 

   |    * result = 'Scarborough, Ontario, Canada'                                                     | 

   |    *                                                                                             | 

   |    * %REPLACE with 4 parameters to delete text:                                                  | 

   |   C                   EVAL      result = %REPLACE('': result: 1:                                 | 

   |   C                                            %SCAN(',': result)+1)                             | 

   |    * result = 'Ontario, Canada'                                                                  | 

   |    *                                                                                             | 

   |    * %REPLACE with 4 parameters to add text to the end of the string:                            | 

   |   C                   EVAL      result = %REPLACE(', ' + %CHAR(date):                            | 

   |   C                                            result:                                           | 

   |   C                                            %LEN(result)+1: 0)                                | 

   |    * result = 'Ontario, Canada, 1997-02-03'                                                      | 

   |    *                                                                                             | 

   |    * %REPLACE with 3 parameters to replace fixed-length text at                                  | 

   |    * specified position:  (fixed1 has fixed-length of 15 chars)                                  | 

   |   C                   EVAL      result = %REPLACE(fixed1: result:                                | 

   |   C                                            %SCAN(',': result)+2)                             | 

   |    * result = 'Ontario, California     -03'                                                      | 

   |    *                                                                                             | 

   |    * %REPLACE with 4 parameters to prefix text at beginning:                                     | 

   |   C                   EVAL      result = %REPLACE('Somewhere else: ':                            | 

   |   C                                            result: 1: 0)                                     | 

   |    * result = 'Somewhere else: Ontario, California     -03'                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 160. %REPLACE Example 


25 %SCAN (Scan for Characters)

     %SCAN(search argument : source string {: start}) 

%SCAN returns the first position of the search argument in the source string, or 0 if it was not found. If the start position is specified, the search begins at the starting position. The result is always the position in the source string even if the starting position is specified. The starting position defaults to 1. 


| The first parameter must be of type character, graphic, or UCS-2 . The second parameter must be the same type as the first parameter. The third parameter, if specified, must be numeric with zero decimal positions. 

When any parameter is variable in length, the values of the other parameters are checked against the current length, not the maximum length. 

The type of the return value is unsigned integer. This built-in function can be used anywhere that an unsigned integer expression is valid. 

Note: Unlike the SCAN operation code, %SCAN cannot return an array containing all occurrences of the search string and its results cannot be tested using the %FOUND built-in function. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D source          S             15A  inz('Dr. Doolittle')                                      | 

   |   D pos             S              5U 0                                                          | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   EVAL      pos = %scan('oo' : source)                                       | 

   |    * After the EVAL, pos = 6 because 'oo' begins at position 6 in                                | 

   |    * 'Dr. Doolittle'.                                                                            | 

   |   C                   EVAL      pos = %scan('D' : source : 2)                                    | 

   |    * After the EVAL, pos = 5 because the first 'D' found starting from                           | 

   |    * position 2 is in position 5.                                                                | 

   |   C                   EVAL      pos = %scan('abc' : source)                                      | 

   |    * After the EVAL, pos = 0 because 'abc' is not found in                                       | 

   |    * 'Dr. Doolittle'.                                                                            | 

   |                                                                                                  | 

   |   C                   EVAL      pos = %scan('Dr.' : source : 2)                                  | 

   |    * After the EVAL, pos = 0 because 'Dr.' is not found in                                       | 

   |    * 'Dr. Doolittle', if the search starts at position 2.                                        | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 161. %SCAN Example 


26 %SIZE (Get Size in Bytes)

     %SIZE(variable) 

     %SIZE(literal) 

     %SIZE(array{:*ALL}) 

     %SIZE(table{:*ALL}) 

     %SIZE(multiple occurrence data structure{:*ALL}) 

%SIZE returns the number of bytes occupied by the constant or field. The argument may be a literal, a named constant, a data structure, a data structure subfield, a field, an array or a table name. It cannot, however, contain an expression. The value returned is in unsigned integer format (type U). 

For a graphic literal, the size is the number of bytes occupied by the graphic characters, not including leading and trailing shift characters. 
| For a hexadecimal or UCS-2 literal, the size returned is half the number 
| of hexadecimal digits in the literal. 

For variable-length fields, %SIZE returns the total number of bytes occupied by the field (two bytes longer than the declared maximum length). 

The length returned for a null-capable field (%SIZE) is always its full length, regardless of the setting of its null indicator. 

If the argument is an array name, table name, or multiple occurrence data structure name, the value returned is the size of one element or occurrence. If *ALL is specified as the second parameter for %SIZE, the value returned is the storage taken up by all elements or occurrences. For a multiple-occurrence data structure containing pointer subfields, the size may be greater than the size of one occurrence times the number of occurrences. The system requires that pointers be placed in storage at addresses evenly divisible by 16. As a result, the length of each occurrence may have to be increased enough to make the length an exact multiple of 16 so that the pointer subfields will be positioned correctly in storage for every occurrence. 

%SIZE may be specified anywhere that a numeric constant is allowed on the definition specification and in an expression in the extended factor 2 field of the calculation specification. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D                                                                                              | 

   |   D arr1            S             10    DIM(4)                                                   | 

   |   D table1          S              5    DIM(20)                                                  | 

   |   D field1          S             10                                                             | 

   |   D field2          S              9B 0                                                          | 

   |   D field3          S              5P 2                                                          | 

   |   D num             S              5P 0                                                          | 

   |   D mds             DS            20    occurs(10)                                               | 

   |   D mds_size        C                   const (%size (mds: *all))                                | 

   |   D mds_ptr         DS            20    OCCURS(10)                                               | 

   |   D   pointer                       *                                                            | 

   |   D vCity           S             40A   VARYING INZ('North York')                                | 

   |   D fCity           S             40A           INZ('North York')                                | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |   C                                                                     Result                   | 

   |   C                   eval      num = %SIZE(field1)                        10                    | 

   |   C                   eval      num = %SIZE('HH')                           2                    | 

   |   C                   eval      num = %SIZE(123.4)                          4                    | 

   |   C                   eval      num = %SIZE(-03.00)                         4                    | 

   |   C                   eval      num = %SIZE(arr1)                          10                    | 

   |   C                   eval      num = %SIZE(arr1:*ALL)                     40                    | 

   |   C                   eval      num = %SIZE(table1)                         5                    | 

   |   C                   eval      num = %SIZE(table1:*ALL)                  100                    | 

   |   C                   eval      num = %SIZE(mds)                           20                    | 

   |   C                   eval      num = %SIZE(mds:*ALL)                     200                    | 

   |   C                   EVAL      num = %SIZE(mds_ptr)                       20                    | 

   |   C                   EVAL      num = %SIZE(mds_ptr:*ALL)                 320                    | 

   |   C                   eval      num = %SIZE(field2)                         4                    | 

   |   C                   eval      num = %SIZE(field3)                         3                    | 

   |   C                   eval      n1 = %SIZE(vCity)                          42                    | 

   |   C                   EVAL      n2 = %SIZE(fCity)                          40                    | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 162. %SIZE Example 


27 %STATUS (Return File or Program Status)

     %STATUS{(file_name)} 

%STATUS returns the most recent value set for the program or file status. %STATUS is set whenever the program status or any file status changes, usually when an error occurs. 

If %STATUS is used without the optional file_name parameter, then it returns the program or file status most recently changed. If a file is specified, the value contained in the INFDS *STATUS field for the specified file is returned. The INFDS does not have to be specified for the file. 

%STATUS starts with a return value of 00000 and is reset to 00000 before any operation with an 'E' extender specified begins. 

%STATUS is best checked immediately after an operation with the 'E' extender or an error indicator specified, or at the beginning of an INFSR or the *PSSR subroutine. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   *...1....+....2....+....3....+....4....+....5....+....6....+....7...+....                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * The 'E' extender indicates that if an error occurs, the error                               | 

   |    * is to be handled as though an error indicator were coded.                                   | 

   |    * The success of the operation can then be checked using the                                  | 

   |    * %ERROR built-in function.  The status associated with the error                             | 

   |    * can be checked using the %STATUS built-in function.                                         | 

   |   C                   EXFMT(E)  INFILE                                                           | 

   |   C                   IF        %ERROR                                                           | 

   |   C                   EXSR      CheckError                                                       | 

   |   C                   ENDIF                                                                      | 

   |   C ...                                                                                          | 

   |    *-------------------------------------------------------------------                          | 

   |    * CheckError: Subroutine to process a file I/O error                                          | 

   |    *-------------------------------------------------------------------                          | 

   |   C     CheckError    BEGSR                                                                      | 

   |   C                   SELECT                                                                     | 

   |   C                   WHEN      %STATUS < 01000                                                  | 

   |    * No error occurred                                                                           | 

   |   C                   WHEN      %STATUS = 01211                                                  | 

   |    * Attempted to read a file that was not open                                                  | 

   |   C                   EXSR      InternalError                                                    | 

   |   C                   WHEN      %STATUS = 01331                                                  | 

   |    * The wait time was exceeded for a READ operation                                             | 

   |   C                   EXSR      TimeOut                                                          | 

   |   C                   WHEN      %STATUS = 01261                                                  | 

   |    * Operation to unacquired device                                                              | 

   |   C                   EXSR      DeviceError                                                      | 

   |   C                   WHEN      %STATUS = 01251                                                  | 

   |    * Permanent I/O error                                                                         | 

   |   C                   EXSR      PermError                                                        | 

   |   C                   OTHER                                                                      | 

   |    * Some other error occurred                                                                   | 

   |   C                   EXSR      FileError                                                        | 

   |   C                   ENDSL                                                                      | 

   |   C                   ENDSR                                                                      | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 163. %STATUS and %ERROR with 'E' Extender 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords++++++++++++++++++++++++++                       | 

   |   D Zero            S              5P 0 INZ(0)                                                   | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq....                    | 

   |    * %STATUS starts with a value of 0                                                            | 

   |    *                                                                                             | 

   |    * The following SCAN operation will cause a branch to the *PSSR                               | 

   |    * because the start position has a value of 0.                                                | 

   |   C     'A'           SCAN      'ABC':Zero    Pos                                                | 

   |   C     BAD_SCAN      TAG                                                                        | 

   |    * The following EXFMT operation has an 'E' extender, so %STATUS will                          | 

   |    * be set to 0 before the operation begins.  Therefore, it is                                  | 

   |    * valid to check %STATUS after the operation.                                                 | 

   |    * Since the 'E' extender was coded, %ERROR can also be used to                                | 

   |    * check if an error occurred.                                                                 | 

   |   C                   EXFMT(E)  REC1                                                             | 

   |   C                   IF        %ERROR                                                           | 

   |   C                   SELECT                                                                     | 

   |   C                   WHEN      %STATUS = 01255                                                  | 

   |   C ...                                                                                          | 

   |   C                   WHEN      %STATUS = 01299                                                  | 

   |   C ...                                                                                          | 

   |    * The following scan operation has an error indicator.  %STATUS will                          | 

   |    * not be set to 0 before the operation begins, but %STATUS can be                             | 

   |    * reasonably checked if the error indicator is on.                                            | 

   |   C     'A'           SCAN      'ABC':Zero    Pos                    10                          | 

   |   C                   IF        *IN10 AND %STATUS = 00100                                        | 

   |   C ...                                                                                          | 

   |                                                                                                  | 

   |    * The following scan operation does not produce an error.                                     | 

   |    * Since there is no 'E' extender %STATUS will not be set to 0,                                | 

   |    * so it would return a value of 00100 from the previous error.                                | 

   |    * Therefore, it is unwise to use %STATUS after an operation that                              | 

   |    * does not have an error indicator or the 'E' extender coded since                            | 

   |    * you cannot be sure that the value pertains to the previous                                  | 

   |    * operation.                                                                                  | 

   |   C     'A'           SCAN      'ABC'         Pos                                                | 

   |   C ...                                                                                          | 

   |   C     *PSSR         BEGSR                                                                      | 

   |    * %STATUS can be used in the *PSSR since an error must have occurred.                         | 

   |   C                   IF        %STATUS = 00100                                                  | 

   |   C                   GOTO      BAD_SCAN                                                         | 

   |   C ...                                                                                          | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 164. %STATUS and %ERROR with 'E' Extender, Error Indicator and *PSSR 

28 %STR (Get or Store Null-Terminated String)

     %STR(basing pointer{: max-length})(right-hand-side) 

     %STR(basing pointer : max-length)(left-hand-side) 


| %STR is used to create or use null-terminated character strings, which are very commonly used in C and C++ applications. 

The first parameter must be a basing-pointer variable. The second parameter, if specified, must be a numeric value with zero decimal 
| positions. If not specified, it defaults to 65535 . 

The first parameter must point to storage that is at least as long as the length given by the second parameter. 

Error conditions: 

2. 
| If the length parameter is not between 1 and 65535 , an error will occur. 

3. If the pointer is not set, an error will occur. 

4. If the storage addressed by the pointer is shorter than indicated by the length parameter, either 

1. An error will occur 

2. Data corruption will occur. 

Subtopics: 

· 28.1 %STR Used to Get Null-Terminated String 

· 28.2 %STR Used to Store Null-Terminated String 

28.1 %STR Used to Get Null-Terminated String

When used on the right-hand side of an expression, this function returns the data pointed to by the first parameter up to but not including the first null character (x'00') found within the length specified. This built-in function can be used anywhere that a character expression is valid. No error will be given at run time if the null terminator is not found within the length specified. In this case, the length of the resulting value is the same as the length specified. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   D String1         S               *                                                            | 

   |   D Fld1            S             10A                                                            | 

   |   C                   EVAL      Fld1 = '<' + %str(String1) + '>'                                 | 

   |    * Assuming that String1 points to '123¬' where '¬' represents the                             | 

   |    * null character, after the EVAL, Fld1 = '<123>     '.                                        | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 165. %STR (right-hand-side) Example 1 

The following is an example of %STR with the second parameter specified. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   D String1         S               *                                                            | 

   |   D Fld1            S             10A                                                            | 

   |   C                   EVAL      Fld1 = '<' + %str(String1 : 2) + '>'                             | 

   |    * Assuming that String1 points to '123¬' where '¬' represents the                             | 

   |    * null character, after the EVAL, Fld1 = '<12>      '.                                        | 

   |    * Since the maximum length read by the operation was 2, the '3' and                           | 

   |    * the '¬' were not considered.                                                                | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 166. %STR (right-hand-side) Example 2 

In this example, the null-terminator is found within the specified maximum length. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   D String1         S               *                                                            | 

   |   D Fld1            S             10A                                                            | 

   |   C                   EVAL      Fld1 = '<' + %str(String1 : 5) + '>'                             | 

   |    * Assuming that String1 points to '123¬' where '¬' represents the                             | 

   |    * null character, after the EVAL, Fld1 = '<123>     '.                                        | 

   |    * Since the maximum length read by the operation was 5, the                                   | 

   |    * null-terminator in position 4 was found so all the data up to                               | 

   |    * the null-terminator was used.                                                               | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 167. %STR (right-hand-side) Example 3 


28.2 %STR Used to Store Null-Terminated String

When used on the left-hand side of an expression, %STR(ptr:length) assigns the value of the right-hand side of the expression to the storage pointed at by the pointer, adding a null-terminating byte at the end. The maximum 
| length that can be specified is 65535 . This means that at most 65534 bytes of the right-hand side can be used, since 1 byte must be reserved for the null-terminator at the end. 

The length indicates the amount of storage that the pointer points to. This length should be greater than the maximum length the right-hand side will have. The pointer must be set to point to storage at least as long as the length parameter. If the length of the right-hand side of the expression is longer than the specified length, the right-hand side value is truncated. 

Note: Data corruption will occur if both of the following are true: 

5. The length parameter is greater than the actual length of data addressed by the pointer. 

6. The length of the right-hand side is greater than or equal to the actual length of data addressed by the pointer. 

If you are dynamically allocating storage for use by %STR, you must keep track of the length that you have allocated. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   D String1         S               *                                                            | 

   |   D Fld1            S             10A                                                            | 

   |    ...                                                                                           | 

   |   C                   EVAL      %str(String1:25) = 'abcdef'                                      | 

   |    * The storage pointed at by String1 now contains 'abcdef¬'                                    | 

   |    * Bytes 8-25 following the null-terminator are unchanged.                                     | 

   |   D String1         S               *                                                            | 

   |   D Fld1            S             10A                                                            | 

   |    ...                                                                                           | 

   |   C                   EVAL      %str(String1 : 4) = 'abcdef'                                     | 

   |    * The storage pointed at by String1 now contains 'abc¬'                                       | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 168. %STR (left-hand-side) Examples 


29 %SUBST (Get Substring)

     %SUBST(string:start{:length}) 

%SUBST returns a portion of argument string. It may also be used as the result of an assignment with the EVAL operation code. 

The start parameter represents the starting position of the substring. 

The length parameter represents the length of the substring. If it is not specified, the length is the length of the string parameter less the start value plus one. 


| The string must be character, graphic, or UCS-2 data. Starting position and length may be any numeric value or numeric expression with zero decimal positions. The starting position must be greater than zero. The length may be greater than or equal to zero. 

When the string parameter is varying length, the values of the other parameters are checked against the current length, not the maximum length. 

When specified as a parameter for a definition specification keyword, the parameters must be literals or named constants representing literals. When specified on a free-form calculation specification, the parameters may be any expression. 

Subtopics: 

· 29.1 %SUBST Used for its Value 

· 29.2 %SUBST Used as the Result of an Assignment 

29.1 %SUBST Used for its Value

%SUBST returns a substring from the contents of the specified string. The 
| string may be any character, graphic, or UCS-2 field or expression. Unindexed arrays are allowed for string, start, and length. The substring begins at the specified starting position in the string and continues for the length specified. If length is not specified then the substring continues to the end of the string. For example: 

     The value of  %subst('Hello World': 5+2) is  'World' 

     The value of  %subst('Hello World':5+2:10-7) is 'Wor' 

     The value of  %subst('abcd' + 'efgh':4:3) is 'def' 


| For graphic or UCS-2 characters the start position and length is consistent with the 2-byte character length (position 3 is the third 2-byte character and length 3 represents 3 2-byte characters to be operated on). 


29.2 %SUBST Used as the Result of an Assignment

When used as the result of an assignment this built-in function refers to certain positions of the argument string. Unindexed arrays are not allowed for start and length. 

The result begins at the specified starting position in the variable and continues for the length specified. If the length is not specified then the string is referenced to its end. If the length refers to characters beyond the end of the string, then a run-time error is issued. 

When %SUBST is used as the result of an assignment, the first parameter must refer to a storage location. That is, the first parameter of the %SUBST operation must be one of the following. 

· Field 

· Data Structure 

· Data Structure Subfield 

· Array Name 

· Array Element 

· Table Element 

Any valid expressions are permitted for the second and third parameters of %SUBST when it appears as the result of an assignment with an EVAL operation. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * In this example, CITY contains 'Toronto, Ontario'                                           | 

   |    * %SUBST returns the value 'Ontario'.                                                         | 

   |    *                                                                                             | 

   |   C     ' '           SCAN      CITY          C                                                  | 

   |   C                   IF        %SUBST(CITY:C+1) = 'Ontario'                                     | 

   |   C                   EVAL      CITYCNT = CITYCNT+1                                              | 

   |   C                   ENDIF                                                                      | 

   |    *                                                                                             | 

   |    * Before the EVAL, A has the value 'abcdefghijklmno'.                                         | 

   |    * After the EVAL A has the value 'ab****ghijklmno'                                            | 

   |    *                                                                                             | 

   |   C                   EVAL      %SUBST(A:3:4) = '****'                                           | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 169. %SUBST Example 


30 %TRIM (Trim Blanks at Edges)

     %TRIM(string) 

%TRIM returns the given string less any leading and trailing blanks. 


| The string can be character, graphic, or UCS-2 data. 

When specified as a parameter for a definition specification keyword, the string parameter must be a constant. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D                                                                                              | 

   |   D LOCATION        S             16A                                                            | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * LOCATION will have the value 'Toronto, Ontario'.                                            | 

   |    *                                                                                             | 

   |   C                   EVAL      LOCATION = %TRIM('  Toronto, Ontario  ')                         | 

   |    *                                                                                             | 

   |    * Name will have the value 'Chris Smith'.                                                     | 

   |    *                                                                                             | 

   |   C                   MOVE(P)   'Chris'       FIRSTNAME        10                                | 

   |   C                   MOVE(P)   'Smith'       LASTNAME         10                                | 

   |   C                   EVAL      NAME =                                                           | 

   |   C                             %TRIM(FIRSTNAME) +' '+ %TRIM(LASTNAME)                           | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 170. %TRIM Example 


31 %TRIML (Trim Leading Blanks)

     %TRIML(string) 

%TRIML returns the given string less any leading blanks. 


| The string can be character, graphic, or UCS-2 data. 

When specified as a parameter for a definition specification keyword, the string parameter must be a constant. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * LOCATION will have the value 'Toronto, Ontario  '.                                          | 

   |    *                                                                                             | 

   |   C                   EVAL      LOCATION = %TRIML('  Toronto, Ontario  ')                        | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 171. %TRIML Example 


32 %TRIMR (Trim Trailing Blanks)

     %TRIMR(string) 

%TRIMR returns the given string less any trailing blanks. 


| The string can be character, graphic, or UCS-2 data. 

When specified as a parameter for a definition specification keyword, the string parameter must be a constant. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++++                      | 

   |   D                                                                                              | 

   |   D LOCATION        S             18A                                                            | 

   |   CL0N01Factor1+++++++Opcode(E)+Factor2+++++++Result++++++++Len++D+HiLoEq..                      | 

   |   CL0N01Factor1+++++++Opcode(E)+Extended-factor2+++++++++++++++++++++++++++                      | 

   |    *                                                                                             | 

   |    * LOCATION will have the value '  Toronto, Ontario'.                                          | 

   |    *                                                                                             | 

   |   C                   EVAL      LOCATION = %TRIMR('  Toronto, Ontario  ')                        | 

   |    *                                                                                             | 

   |    * Name will have the value 'Chris Smith'.                                                     | 

   |    *                                                                                             | 

   |   C                   MOVEL(P)  'Chris'       FIRSTNAME        10                                | 

   |   C                   MOVEL(P)  'Smith'       LASTNAME         10                                | 

   |   C                   EVAL      NAME =                                                           | 

   |   C                             %TRIMR(FIRSTNAME) +' '+ %TRIMR(LASTNAME)                         | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 172. %TRIMR Example 


33 %UCS2 (Convert to UCS-2 Value)


| %UCS2 converts the value of the expression from character, graphic, or 
| UCS-2 and returns a UCS-2 value. The result is varying length if the 
| parameter is varying length, or if the parameter is single-byte character. 


| The second parameter, ccsid, is optional and indicates the CCSID of the 
| resulting expression. The CCSID defaults to 13488. 


| If the parameter is a constant, the conversion will be done at compile 
| time. 


| If the conversion results in substitution characters, a warning message is 
| issued at compile time. At run time, status 00050 is set and no error 
| message is issued. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

 | |   HKeywords++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++                        | 

 | |   H CCSID(*UCS2 : 13488)                                                                         | 

 | |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++                        | 

 | |   D  char           S              5A   INZ('abcde')                                             | 

 | |   D  graph          S              2G   INZ(G'oAABBi')                                           | 

 | |    * The %UCS2 built-in function is used to initialize a UCS-2 field.                            | 

 | |   D  ufield         S             10C   INZ(%UCS2('abcdefghij'))                                 | 

 | |   D  ufield2        S              1C   CCSID(61952) INZ(*LOVAL)                                 | 

 | |   D  isLess                        1N                                                            | 

 | |   D  proc           PR                                                                           | 

 | |   D    uparm                       2G   CCSID(13488) CONST                                       | 

 | |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++                        | 

 | |   C                   EVAL      ufield = %UCS2(char) +  %UCS2(graph)                             | 

 | |    * ufield now has 7 UCS-2 characters representing                                              | 

 | |    * 'a.b.c.d.e.AABB' where 'x.' represents the UCS-2 form of 'x'                                | 

   |                                                                                                  | 

 | |   C                   EVAL      isLess = ufield < %UCS2(ufield2:13488)                           | 

 | |    * The result of the %UCS2 built-in function is the value of                                   | 

 | |    * ufield2, converted from CCSID 61952 to CCSID 13488                                          | 

 | |    * for the comparison.                                                                         | 

   |                                                                                                  | 

 | |   C                   EVAL      ufield = ufield2                                                 | 

 | |    * The value of ufield2 is converted from CCSID 61952 to                                       | 

 | |    * CCSID 13488 and stored in ufield.                                                           | 

 | |    * This conversion is handled implicitly by the compiler.                                      | 

   |                                                                                                  | 

 | |   C                   CALLP     proc(ufield2)                                                    | 

 | |    * The value of ufield2 is converted to CCSID 13488                                            | 

 | |    * implicitly, as part of passing the parameter by constant reference.                         | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

 | Figure 173. %UCS2 Examples 


34 %UNS (Convert to Unsigned Format)

     %UNS(numeric expression) 

%UNS converts the value of the numeric expression to unsigned format. Any decimal digits are truncated. %UNS can be used to truncate the decimal positions from a float or decimal value allowing it to be used as an array index. 

Subtopics: 

· 34.1 %UNSH (Convert to Unsigned Format with Half Adjust) 

34.1 %UNSH (Convert to Unsigned Format with Half Adjust)

     %UNSH(numeric expression) 

%UNSH is like %UNS except that if the numeric expression is a decimal or a float value, half adjust is applied to the value of the numeric expression when converting to unsigned type. No message is issued if half adjust cannot be performed. 

    __________________________________________________________________________________________________  

   |                                                                                                  | 

   |                                                                                                  | 

   |   DName+++++++++++ETDsFrom+++To/L+++IDc.Keywords+++++++++++++++++++++++++                        | 

   |   D p7              s              7p 3 inz (8236.567)                                           | 

   |   D s9              s              9s 5 inz (23.73442)                                           | 

   |   D f8              s              8f   inz (173.789)                                            | 

   |   D result1         s             15p 5                                                          | 

   |   D result2         s             15p 5                                                          | 

   |   D result3         s             15p 5                                                          | 

   |   D array           s              1a   dim (200)                                                | 

   |   D a               s              1a                                                            | 

   |   CL0N01Factor1+++++++Opcode&ExtExtended-factor2+++++++++++++++++++++++++++                      | 

   |   C                   eval      result1 = %uns (p7) + 0.1234                                     | 

   |   C                   eval      result2 = %uns (s9)                                              | 

   |   C                   eval      result3 = %unsh (f8)                                             | 

   |    * The value of "result1" is now 8236.12340.                                                   | 

   |    * The value of "result2" is now   23.00000                                                    | 

   |    * The value of "result3" is now  174.00000.                                                   | 

   |   C                   eval      a = array (%unsh (f8))                                           | 

   |    * %UNS and %UNSH can be used as array indexes                                                 | 

   |                                                                                                  | 

   |                                                                                                  | 

   |__________________________________________________________________________________________________| 

   Figure 174. %UNS and %UNSH Example 


35 %XFOOT (Sum Array Expression Elements)

 |   %XFOOT(array-expression) 


| %XFOOT results in the sum of all elements of the specified numeric array 
| expression. 


| The precision of the result is the minimum that can hold the result of 
| adding together all array elements, up to a maximum of 30 digits. The 
| number of decimal places in the result is always the same as the decimal 
| places of the array expression. 


| For example, if ARR is an array of 500 elements of precision (17,4), the 
| result of %XFOOT(ARR) is (20,4). 


| For %XFOOT(X) where X has precision (m,n), the following table shows the 
| precision of the result based on the number of elements of X: 

 |   Elements of X      Precision of %XFOOT(X) 

 |   1                  (m,n) 

 |   2-10               (m+1,n) 

 |   11-100             (m+2,n) 

 |   101-1000           (m+3,n) 

 |   1001-10000         (m+4,n) 

 |   10001-32767        (m+5,n) 


| Normal rules for array expressions apply. For example, if ARR1 has 10 
| elements and ARR2 has 20 elements, %XFOOT(ARR1+ARR2) results in the sum of 
| the first 10 elements of ARR1+ARR2. 


| This built-in function is similar to the XFOOT operation, except that 
| float arrays are summed like all other types, beginning from index 1 on 
| up. 

